Pensieve header: Mathematica notebook for A Perturbed Alexander Invariant.

In[∘]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\APAI"];
```

The Program

tex

Two of the main reasons we like ρ_1 is that it is very easy to implement and even an unsophisticated implementation runs very fast. To highlight these points we include a full implementation here, a step-by-step run-through, and a demo run. We write in Mathematica~\cite{Wolfram:Mathematica}, and you can find the notebook displayed here at~\cite[APAI.nb]{Self}.

We start by loading the library \verb$KnotTheory`$~\cite{Bar-NatanMorrison:KnotTheory} (it is used here only for the list of knots that it contains, and to compute other invariants for comparisons). We also load a minor conversion routine~\cite[Rot.nb / Rot.m]{Self} whose internal workings are irrelevant here.

pdf

In[∘]:=
```
Once[<< KnotTheory`; << Rot.m];
```

pdf

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

pdf

```
"Loading Rot.m from http://drorbn.net/APAI to compute rotation numbers.
```

tex

\ifpub{}{\needspace{50mm}}
\subsection{The Program} This done, here is the full ρ_1 program:

pdf

In[∘]:=
```
R₁[s_, i_, j_] := s (g_{ji} (g_{j⁺,j} + g_{j,j⁺} − g_{ij}) − g_{ii} (g_{j,j⁺} − 1) − 1 / 2);
ρ[K_] := ρ[K] = Module[{Cs, φ, n, A, s, i, j, k, Δ, G, ρ1},
    {Cs, φ} = Rot[K]; n = Length[Cs];
    A = IdentityMatrix[2 n + 1];
    Cases[Cs, {s_, i_, j_} :→ (A[[{i, j}, {i + 1, j + 1}]] += ( −T^s  T^s − 1
                                                                   0      −1   ))];
    Δ = T^((−Total[φ]−Total[Cs[[All,1]]])/2) Det[A];
    G = Inverse[A];
    ρ1 = Sum_{k=1}^{n} R₁ @@ Cs[[k]] − Sum_{k=1}^{2 n} φ[[k]] (g_{kk} − 1 / 2);
    Factor@{Δ, Δ^2 ρ1 /. α_⁺ :→ α + 1 /. g_{α_,β_} :→ G[[α, β]]}];
```

tex

The program uses mostly the same symbols as the text, so even without any knowledge of Mathematica, the reader should be able to recognize at least formulas~\eqref{eq:A}, \eqref{eq:Delta}, and~\eqref{eq:rho1} within it. As a further hint we add that the variable \verb$Cs$ ends up storing the list of crossings in a knot K, where each crossing is stored as a triple (s,i,j), where s, i, and j have the same meaning as in~\eqref{eq:A}. The conversion routine \verb$Rot$ automatically produces \verb$Cs$, as well as a list φ of rotation numbers, given any other knot presentation known to the package \verb$KnotTheory`$.

tex

Note that the program outputs the ordered pair (Δ,ρ_1). The Alexander polynomial Δ is anyway computed internally, and we consider the aggregate (Δ,ρ_1) as more interesting than any of its pieces by itself.

Step-by-step Run-Through

tex

\subsection{A Step-by-Step Run-Through} We start by setting K to be the knot diagram on page~1 using the \verb$PD$ notation of \verb$KnotTheory`$~\cite{Bar-NatanMorrison:KnotTheory}. We then print \verb$Rot[K]$, which is a list of crossings followed by a list of rotation numbers:

pdf

```
In[•]:= K = PD[X[4, 2, 5, 1], X[2, 6, 3, 5], X[6, 4, 7, 3]];
Rot[K]
```

Out[•]=
pdf

```
{{{1, 1, 4}, {1, 5, 2}, {1, 3, 6}}, {0, 0, 0, -1, 0, 0}}
```

tex

Next we set \verb$Cs$ and φ to be the list of crossings, and the list of rotation numbers, respectively.
\ifpub{}{\needspace{20mm}}

pdf

```
In[•]:= {Cs, φ} = Rot[K]
```

Out[•]=
pdf

```
{{{1, 1, 4}, {1, 5, 2}, {1, 3, 6}}, {0, 0, 0, -1, 0, 0}}
```

tex

We set \verb$n$ to be the number of crossings, \verb$A$ to be the $(2n+1)$-dimensional identity matrix, and then we iterate over \verb$c$ in \verb$Cs$, adding a block as in~\eqref{eq:A} for each crossing.

pdf

```
In[•]:= n = Length[Cs];
A = IdentityMatrix[2 n + 1];
Cases[Cs, {s_, i_, j_} :> (A[[{i, j}, {i + 1, j + 1}]] += ( -T^s  T^s - 1 ; 0  -1 ))];
```

$$\texttt{Cases}\left[\texttt{Cs}, \{s\_, i\_, j\_\} \mapsto \left(\texttt{A}[\![\{i,j\},\{i+1,j+1\}]\!] \mathrel{+}= \begin{pmatrix} -T^s & T^s-1 \\ 0 & -1 \end{pmatrix}\right)\right];$$

tex

\ifpub{}{\needspace{30mm}}
Here's what \verb$A$ comes out to be:

pdf

In[∘]:= A // MatrixForm

Out[∘]//MatrixForm=
pdf

$$\begin{pmatrix} 1 & -T & 0 & 0 & -1+T & 0 & 0 \\ 0 & 1 & -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & -T & 0 & 0 & -1+T \\ 0 & 0 & 0 & 1 & -1 & 0 & 0 \\ 0 & 0 & -1+T & 0 & 1 & -T & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & -1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

Here's the same, in TeXForm:

In[∘]:= A // MatrixForm // TeXForm

Out[∘]//TeXForm=

```
\left(
\begin{array}{ccccccc}
 1 & -T & 0 & 0 & T-1 & 0 & 0 \\
 0 & 1 & -1 & 0 & 0 & 0 & 0 \\
 0 & 0 & 1 & -T & 0 & 0 & T-1 \\
 0 & 0 & 0 & 1 & -1 & 0 & 0 \\
 0 & 0 & T-1 & 0 & 1 & -T & 0 \\
 0 & 0 & 0 & 0 & 0 & 1 & -1 \\
 0 & 0 & 0 & 0 & 0 & 0 & 1 \\
\end{array}
\right)
```

tex

We set \$\Delta\$ to be the determinant of \verb\$A\$, with a correction as in~\eqref{eq:Delta}. So \$\Delta\$ is the Alexander polynomial of \$K\$.

In[∘]:= Det[A]

Out[∘]=

$1 - T + T^2$

pdf

In[∘]:= Δ = T^(-Total[φ]-Total[Cs⟦All,1⟧])/2 Det[A]

Out[∘]=
pdf

$$\frac{1 - T + T^2}{T}$$

tex

\ifpub{}{\needspace{30mm}}
\verb\$G\$ is now the \verb\$Inverse\$ of \verb\$A\$:

pdf

In[∘]:=
```
G = Inverse[A];
G // MatrixForm
```

Out[∘]//MatrixForm=
pdf

$$\begin{pmatrix}
1 & \frac{T-T^2+T^3}{1-T+T^2} & 1 & \frac{T-T^2+T^3}{1-T+T^2} & 1 & \frac{T-T^2+T^3}{1-T+T^2} & 1 \\
0 & 1 & \frac{1}{1-T+T^2} & \frac{T}{1-T+T^2} & \frac{T}{1-T+T^2} & \frac{T^2}{1-T+T^2} & 1 \\
0 & 0 & \frac{1}{1-T+T^2} & \frac{T}{1-T+T^2} & \frac{T}{1-T+T^2} & \frac{T^2}{1-T+T^2} & 1 \\
0 & 0 & \frac{1-T}{1-T+T^2} & \frac{1}{1-T+T^2} & \frac{1}{1-T+T^2} & \frac{T}{1-T+T^2} & 1 \\
0 & 0 & \frac{1-T}{1-T+T^2} & \frac{T-T^2}{1-T+T^2} & \frac{1}{1-T+T^2} & \frac{T}{1-T+T^2} & 1 \\
0 & 0 & 0 & 0 & 0 & 1 & 1 \\
0 & 0 & 0 & 0 & 0 & 0 & 1
\end{pmatrix}$$

Here’s the same, in TeXForm:

In[∘]:=
```
G // MatrixForm // TeXForm
```

Out[∘]//TeXForm=
```
\left(
\begin{array}{ccccccc}
 1 & \frac{T^3-T^2+T}{T^2-T+1} & 1 & \frac{T^3-T^2+T}{T^2-T+1} & 1 &
   \frac{T^3-T^2+T}{T^2-T+1} & 1 \\
 0 & 1 & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T}{T^2-T+1} &
   \frac{T^2}{T^2-T+1} & 1 \\
 0 & 0 & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T}{T^2-T+1} &
   \frac{T^2}{T^2-T+1} & 1 \\
 0 & 0 & \frac{1-T}{T^2-T+1} & \frac{1}{T^2-T+1} & \frac{1}{T^2-T+1} &
   \frac{T}{T^2-T+1} & 1 \\
 0 & 0 & \frac{1-T}{T^2-T+1} & \frac{T-T^2}{T^2-T+1} & \frac{1}{T^2-T+1} &
   \frac{T}{T^2-T+1} & 1 \\
 0 & 0 & 0 & 0 & 0 & 1 & 1 \\
 0 & 0 & 0 & 0 & 0 & 0 & 1 \\
\end{array}
\right)
```

tex

\ifpub{}{\needspace{30mm}}

It remains to blindly follow the two parts of Equation~\eqref{eq:rho1}:

pdf

In[∘]:= $\rho 1 = \sum_{k=1}^{n} R_1 \,@@\, Cs[\![k]\!] - \sum_{k=1}^{2n} \varphi[\![k]\!]\,(g_{kk} - 1/2)$

Out[∘]=
pdf

$$-2 + g_{4,4} - g_{1,1}\,(-1 + g_{4,4^+}) - (-1 + g_{2,2^+})\,g_{5,5} - g_{3,3}\,(-1 + g_{6,6^+}) + g_{2,5}\,(g_{2,2^+} - g_{5,2} + g_{2^+,2}) + g_{4,1}\,(-g_{1,4} + g_{4,4^+} + g_{4^+,4}) + g_{6,3}\,(-g_{3,6} + g_{6,6^+} + g_{6^+,6})$$

tex

We replace each ${\mathtt g}_{\alpha\beta}$ with the appropriate entry of \verb$G$:

pdf

In[∘]:= $\Delta^2\,\rho 1$ /. $\alpha\_^+ \mapsto \alpha+1$ /. $g_{\alpha\_,\beta\_} \mapsto G[\![\alpha,\beta]\!]$

Out[∘]=
pdf

$$\frac{\left(1-T+T^2\right)^2\left(-1+\frac{T}{\left(1-T+T^2\right)^2}-\frac{-1+\frac{1}{1-T+T^2}}{1-T+T^2}\right)}{T^2}$$

tex

Finally, we output both Δ and ρ_1. We factor them just to put them in a nicer form:

pdf

In[∘]:= Factor@{Δ, $\Delta^2\,\rho 1$ /. $\alpha\_^+ \mapsto \alpha+1$ /. $g_{\alpha\_,\beta\_} \mapsto G[\![\alpha,\beta]\!]$}

Out[∘]=
pdf

$$\left\{\frac{1-T+T^2}{T},\ -\frac{(-1+T)^2\left(1+T^2\right)}{T^2}\right\}$$

A Demo Run

tex

\subsection{A Demo Run} \label{ssec:Demo} Here are Δ and ρ_1 of all the knots with up to 6 crossings (a table up to 10 crossings is printed at~\cite{PG}:

pdf

In[∘]:= TableForm[Table[Join[{K}, ρ[K]], {K, AllKnots[{3, 6}]}], TableAlignments → Center]

pdf

KnotTheory: Loading precomputed data in PD4Knots`.

Out[∘]//TableForm=
pdf

Knot[3, 1]	$\frac{1-T+T^2}{T}$	$\frac{(-1+T)^2\left(1+T^2\right)}{T^2}$
Knot[4, 1]	$-\frac{1-3T+T^2}{T}$	0
Knot[5, 1]	$\frac{1-T+T^2-T^3+T^4}{T^2}$	$\frac{(-1+T)^2\left(1+T^2\right)\left(2+T^2+2T^4\right)}{T^4}$
Knot[5, 2]	$\frac{2-3T+2T^2}{T}$	$\frac{(-1+T)^2\left(5-4T+5T^2\right)}{T^2}$
Knot[6, 1]	$-\frac{(-2+T)(-1+2T)}{T}$	$\frac{(-1+T)^2\left(1-4T+T^2\right)}{T^2}$
Knot[6, 2]	$-\frac{1-3T+3T^2-3T^3+T^4}{T^2}$	$\frac{(-1+T)^2\left(1-4T+4T^2-4T^3+4T^4-4T^5+T^6\right)}{T^4}$
Knot[6, 3]	$\frac{1-3T+5T^2-3T^3+T^4}{T^2}$	0

tex

Some comments are in order:
\begin{itemize}
\item If $\bar{K}$ is the mirror of a knot K, then $\rho_1(\bar{K})(T) = -\rho_1(K)(T^{-1})$. Indeed in~\eqref{eq:rho1} both $R_1(c)$ and φ_k flip sign under reflection in a plane perpendicular to the plane of the knot diagram, and the matrix A and hence also all the $g_{\alpha\beta}$'s are the same except for the substitution $T\to T^{-1}$.
\item ρ_1 seems to always be divisible by $(T-1)^2$ and seems to always be palindromic

(\$\rho_1(T)=\rho_1(T^{-1})\$). We are not sure why this is so.
\item The last properties taken together would imply that \$\rho_1\$ vanishes on amphicheiral knots, such as \$4_1\$ and \$6_3\$ above.
\end{itemize}

tex

\begin{figure}
\[\resizebox{\ifpub{\linewidth}{6in}}{!}{\input{figs/GST48-Marked.pdf_t}} \]
\caption{A 48-crossing knot from~\cite{GompfScharlemannThompson:Counterexample}.} \label{-fig:GST48}
\end{figure}
Next is one of our favourites, a knot from~\cite{GompfScharlemannThompson:Counterexample} (see Figure~\ref{fig:GST48}), which is a potential counterexample to the ribbon\$=\$slice conjecture~\cite{-Fox:Problems}. It takes about two minutes to compute \$\rho_1\$ for this 48 crossing knot (note that Mathematica prints \verb\$Timing\$ information is seconds, and that this information is highly dependent on the CPU used, how loaded it is, and even on its temperature at the time of the computation):

pdf

In[∘]:=

$$\text{Timing}@\rho\big[\text{EPD}\big[X_{14,1}, \bar{X}_{2,29}, X_{3,40}, X_{43,4}, \bar{X}_{26,5}, X_{6,95}, X_{96,7}, X_{13,8}, \bar{X}_{9,28}, X_{10,41}, X_{42,11}, \bar{X}_{27,12}, X_{30,15}, \bar{X}_{16,61}, \bar{X}_{17,72}, \bar{X}_{18,83}, X_{19,34}, \bar{X}_{89,20}, \bar{X}_{21,92}, \bar{X}_{79,22}, \bar{X}_{68,23}, \bar{X}_{57,24}, \bar{X}_{25,56}, X_{62,31}, X_{73,32}, X_{84,33}, \bar{X}_{50,35}, X_{36,81}, X_{37,70}, X_{38,59}, \bar{X}_{39,54}, X_{44,55}, X_{58,45}, X_{69,46}, X_{80,47}, X_{48,91}, X_{90,49}, X_{51,82}, X_{52,71}, X_{53,60}, \bar{X}_{63,74}, \bar{X}_{64,85}, \bar{X}_{76,65}, \bar{X}_{87,66}, \bar{X}_{67,94}, \bar{X}_{75,86}, \bar{X}_{88,77}, \bar{X}_{78,93}\big]\big]$$

Out[∘]=
pdf

$$\Big\{158.625, \Big\{-\frac{(-1+2T-T^2-T^3+2T^4-T^5+T^8)(-1+T^3-2T^4+T^5+T^6-2T^7+T^8)}{T^8},$$
$$\frac{1}{T^{16}}(-1+T)^2(5-18T+33T^2-32T^3+2T^4+42T^5-62T^6-8T^7+166T^8-242T^9+108T^{10}+132T^{11}-226T^{12}+148T^{13}-11T^{14}-36T^{15}-11T^{16}+148T^{17}-226T^{18}+132T^{19}+108T^{20}-242T^{21}+166T^{22}-8T^{23}-62T^{24}+42T^{25}+2T^{26}-32T^{27}+33T^{28}-18T^{29}+5T^{30})\Big\}\Big\}$$

tex

\subsection{The Separation Power of \$\rho_1\$} Let us check how powerful \$\rho_1\$ is on knots with up to 12 crossings:

pdf

```
{NumberOfKnots[{3, 12}],
 Length@Union@Table[ρ[K], {K, AllKnots[{3, 12}]}],
 Length@Union@Table[{HOMFLYPT[K], Kh[K]}, {K, AllKnots[{3, 12}]}]}
```

Out[∘]=
pdf

```
{2977, 2882, 2785}
```

tex

So the pair \$(\Delta,\rho_1)\$ attains 2,882 distinct values on the 2,977 prime knots with up to 12 crossings (a deficit of 95), whereas the pair \$(H,Kh)=\$ (HOMFLYPT polynomial, Khovanov Homology) attains only 2,785 distinct values on the same knots (a deficit of 192).

tex

In our spare time we computed all of these invariants on all the prime knots with up to 14 crossings. On these 59,937 knots the pair (Δ,ρ_1) attains 53,684 distinct values (a deficit of 6,253) whereas the pair (H,Kh) attains only 49,149 distinct values on the same knots (a deficit of 10,788).

tex

Hence the pair (Δ,ρ_1), computable in polynomial time by simple programs, seems stronger than the pair (H,Kh), which is more difficult to program and (for all we know) cannot be computed in polynomial time. We are not aware of another poly-time invariant as strong as the pair (Δ,ρ_1).

Genus

\subsection{ρ_1 and the Knot Genus} \label{ssec:Genus} Here are the 25 knots with up to 12 crossings for which the genus bound coming from ρ_1 is better than the genus bound coming from the Alexander polynomial (see Theorem~\ref{thm:Genus}):

In[∘]:=
```
Breadth[p_] := Exponent[p, T, Max] - Exponent[p, T, Min];
Table[If[Breadth[ρ[K]〚1〛] / 2 < Breadth[ρ[K]〚2〛] / 4, K, Nothing],
  {K, AllKnots[{3, 12}]}] /. Knot[n_, NonAlternating, k_] ⧴ n_{"n"<>ToString[k]}
```

Out[∘]=

{11_{n34}, 11_{n42}, 11_{n67}, 11_{n97}, 12_{n23}, 12_{n31}, 12_{n51}, 12_{n124}, 12_{n129}, 12_{n256}, 12_{n257}, 12_{n264}, 12_{n267}, 12_{n268}, 12_{n293}, 12_{n313}, 12_{n321}, 12_{n411}, 12_{n430}, 12_{n457}, 12_{n519}, 12_{n665}, 12_{n750}, 12_{n812}, 12_{n830}}

The g-Rules

exec

```
nb2tex$TeXFileName = "gRules.tex";
```

pdf

In[∘]:=

$$\delta_{i\_,j\_} := \text{If}[i === j, 1, 0];$$
$$\text{gRules}_{s\_,i\_,j\_} := \{g_{i\beta\_} \mapsto \delta_{i\beta} + T^s\, g_{i^+,\beta} + (1 - T^s)\, g_{j^+,\beta},\ g_{j\beta\_} \mapsto \delta_{j\beta} + g_{j^+,\beta},$$
$$g_{\alpha\_,i} \mapsto T^{-s}\,(g_{\alpha,i^+} - \delta_{\alpha,i^+}),\ g_{\alpha\_j} \mapsto g_{\alpha,j^+} - (1 - T^s)\, g_{\alpha i} - \delta_{\alpha,j^+}\}$$
$$(\alpha\_^+)^+ := \alpha^{"++"};\quad \text{(* this is for cosmetic reasons only *)}$$

Invariance Under R3

exec

```
nb2tex$TeXFileName = "Invariance-R3-Short.tex";
```

pdf

In[∘]:=

$$\text{lhs} = R_1[1, j, k] + R_1[1, i, k^+] + R_1[1, i^+, j^+] \text{ //. } \text{gRules}_{1,j,k} \cup \text{gRules}_{1,i,k^+} \cup \text{gRules}_{1,i^+,j^+};$$
$$\text{rhs} = R_1[1, i, j] + R_1[1, i^+, k] + R_1[1, j^+, k^+] \text{ //. } \text{gRules}_{1,i,j} \cup \text{gRules}_{1,i^+,k} \cup \text{gRules}_{1,j^+,k^+};$$
$$\text{Simplify}[\text{lhs} == \text{rhs}]$$

Out[∘]=
pdf

True

exec

```
nb2tex$TeXFileName = "Invariance-R3.tex";
```

tex

\ifpub{}{\needspace{30mm}}

pdf

In[∘]:=
```
lhs = Simplify[
  R₁[1, j, k] + R₁[1, i, k⁺] + R₁[1, i⁺, j⁺] //. gRules_{1,j,k} ⋃ gRules_{1,i,k⁺} ⋃ gRules_{1,i⁺,j⁺}]
```

Out[∘]=
pdf

$$-\frac{1}{2T^2}\Big(-2(-1+T)\,T\,g^2_{j^{++},i^{++}}+2g_{j^{++},i^{++}}\big(T^2+T^2 g_{i^{++},j^{++}}-2T^2 g_{j^{++},j^{++}}+g_{k^{++},i^{++}}-2T g_{k^{++},i^{++}}+T^2 g_{k^{++},i^{++}}-T g_{k^{++},j^{++}}+T^2 g_{k^{++},j^{++}}\big)+2g_{i^{++},i^{++}}\big(-2T^2+(-1+T)\,T g_{j^{++},i^{++}}+T^2 g_{j^{++},j^{++}}-g_{k^{++},i^{++}}+T g_{k^{++},i^{++}}+T^2 g_{k^{++},k^{++}}\big)+T\big(3T-2(-1+T)\,g^2_{k^{++},i^{++}}+2T g_{k^{++},j^{++}}+2T g_{j^{++},k^{++}} g_{k^{++},j^{++}}+2g^2_{k^{++},j^{++}}-2T g^2_{k^{++},j^{++}}+2g_{j^{++},j^{++}}\big((-1+T)\,g_{k^{++},i^{++}}+(-1+T)\,g_{k^{++},j^{++}}+T(-1+g_{k^{++},k^{++}})\big)-4T g_{k^{++},j^{++}} g_{k^{++},k^{++}}+2g_{k^{++},i^{++}}\big(T+T g_{i^{++},k^{++}}-2(-1+T)\,g_{k^{++},j^{++}}-2T g_{k^{++},k^{++}}\big)\big)\Big)$$

pdf

In[∘]:=
```
rhs = Simplify[
  R₁[1, i, j] + R₁[1, i⁺, k] + R₁[1, j⁺, k⁺] //. gRules_{1,i,j} ⋃ gRules_{1,i⁺,k} ⋃ gRules_{1,j⁺,k⁺}]
```

Out[∘]=
pdf

$$-\frac{1}{2T^2}\Big(-2(-1+T)\,T\,g^2_{j^{++},i^{++}}+2g_{j^{++},i^{++}}\big(T^2+T^2 g_{i^{++},j^{++}}-2T^2 g_{j^{++},j^{++}}+g_{k^{++},i^{++}}-2T g_{k^{++},i^{++}}+T^2 g_{k^{++},i^{++}}-T g_{k^{++},j^{++}}+T^2 g_{k^{++},j^{++}}\big)+2g_{i^{++},i^{++}}\big(-2T^2+(-1+T)\,T g_{j^{++},i^{++}}+T^2 g_{j^{++},j^{++}}-g_{k^{++},i^{++}}+T g_{k^{++},i^{++}}+T^2 g_{k^{++},k^{++}}\big)+T\big(3T-2(-1+T)\,g^2_{k^{++},i^{++}}+2T g_{k^{++},j^{++}}+2T g_{j^{++},k^{++}} g_{k^{++},j^{++}}+2g^2_{k^{++},j^{++}}-2T g^2_{k^{++},j^{++}}+2g_{j^{++},j^{++}}\big((-1+T)\,g_{k^{++},i^{++}}+(-1+T)\,g_{k^{++},j^{++}}+T(-1+g_{k^{++},k^{++}})\big)-4T g_{k^{++},j^{++}} g_{k^{++},k^{++}}+2g_{k^{++},i^{++}}\big(T+T g_{i^{++},k^{++}}-2(-1+T)\,g_{k^{++},j^{++}}-2T g_{k^{++},k^{++}}\big)\big)\Big)$$

pdf

In[∘]:=
```
lhs == rhs
```

Out[∘]=
pdf

True

Invariance Under R2c

exec

```
nb2tex$TeXFileName = "Invariance-R2c.tex";
```

pdf

In[∘]:=
```
Simplify[R1[-1, i, j+] + R1[1, i+, j] - (g_{j+,j+} - 1 / 2)]
lhs = Simplify[R1[-1, i, j+] + R1[1, i+, j] - (g_{j+,j+} - 1 / 2) //. gRules_{-1,i,j+} ⋃ gRules_{1,i+,j}]
```

Out[∘]=
pdf

$$\frac{1}{2} - (-1 + g_{j,j^+})\, g_{i^+,i^+} + g_{j,i^+}\,(g_{j,j^+} - g_{i^+,j} + g_{j^+,j}) + g_{i,i}\,(-1 + g_{j^+,j^{++}}) - g_{j^+,i}\,(-g_{i,j^+} + g_{j^{++},j^+} + g_{j^+,j^{++}}) - g_{j^+,j^+}$$

Out[∘]=
pdf

$$\frac{1}{2} - g_{j^{++},j^{++}}$$

Invariance Under R1l

exec

```
nb2tex$TeXFileName = "Invariance-R1l.tex";
```

pdf

In[∘]:=
```
lhs1 = R1[1, i+, i] - (g_{i+,i+} - 1 / 2)
lhs2 = lhs1 //. {g_{i+,β_} ⧴ T^-1 δ_{i+,β} + g_{i++,β}, g_{i,β_} ⧴ δ_{i,β} + g_{i+,β}}
Simplify[lhs2]
```

Out[∘]=
pdf

$$g_{i,i^+}^2 - g_{i^+,i^+} - (-1 + g_{i,i^+})\, g_{i^+,i^+}$$

Out[∘]=
pdf

$$-\frac{1}{T} - g_{i^{++},i^+} - \left(-1 + \frac{1}{T} + g_{i^{++},i^+}\right)\left(\frac{1}{T} + g_{i^{++},i^+}\right) + \left(\frac{1}{T} + g_{i^{++},i^+}\right)^2$$

Out[∘]=
pdf

0

R1r, R2b, and Sw$^+$.

exec

```
nb2tex$TeXFileName = "Invariance-Rest.tex";
```

pdf

```
Simplify[R1[1, i, i+] + (g_{i+,i+} - 1 / 2) //. {   (* R1r *)
    g_{iβ_} ⧴ δ_{iβ} + T g_{i+,β} + (1 - T) g_{i++,β}, g_{i+,β_} ⧴ δ_{i+,β} + g_{i++,β},
    g_{α_,i} ⧴ T^-1 (g_{α,i+} - δ_{α,i+}), g_{α_,i+} ⧴ T g_{α,i++} - (1 - T) δ_{α,i+} - T δ_{α,i++} }]
```

Out[∘]=
pdf

0

tex

\noindent(Note that the version of the g-rules we used above easily follows from~\eqref{eq:-CounterRules}).

pdf

```
Simplify[R₁[1, i, j] + R₁[-1, i⁺, j⁺] //. gRules_{1,i,j} ⋃ gRules_{-1,i⁺,j⁺}] (* R2b *)
```

Out[∘]=
pdf

0

pdf

```
(g_{i,i} - 1 / 2) + (g_{j,j} - 1 / 2) - (g_{i⁺,i⁺} - 1 / 2) - (g_{j⁺,j⁺} - 1 / 2) //. gRules_{1,i,j} (* Sw⁺ *)
```

Out[∘]=
pdf

0

In[∘]:=
```
Length[AllKnots[{3, 13}]]
```

Out[∘]=

12 965

In[∘]:=
```
Monitor[Timing[Tallyρ13 = Tally[Last /@ Tally@Table[ρ[K], {K, AllKnots[{3, 13}]}]]], K]
```

KnotTheory: Loading precomputed data in DTCode4KnotsTo11`.

KnotTheory: The GaussCode to PD conversion was written by Siddarth Sankaran at the University of Toronto in the summer of 2005.

KnotTheory: Loading precomputed data in KnotTheory/12A.dts.

KnotTheory: Loading precomputed data in KnotTheory/12N.dts.

General: Further output of KnotTheory::loading will be suppressed during this calculation.

Out[∘]=

{20 270.5, {{1, 11 140}, {2, 809}, {4, 33}, {3, 23}, {6, 1}}}

In[∘]:=
```
Total[Times @@@ Rest[Tallyρ13]]
```

Out[∘]=

1825

In[∘]:=
```
Monitor[Timing[TallyHKh13 = Tally[Last /@
    Tally@Table[{Kh[PD@K][q, t], HOMFLYPT[PD@K][a, z]}, {K, AllKnots[{3, 13}]}]]], K]
```

Out[∘]=

{950., {{1, 9714}, {2, 1269}, {3, 150}, {4, 47}, {5, 10}, {6, 3}, {7, 1}}}

In[∘]:=
```
Total[Times @@@ Rest[TallyHKh13]]
```

Out[∘]=

3251

In[∘]:=
```
{NumberOfKnots[14, Alternating], NumberOfKnots[14, NonAlternating]}
```

Out[∘]=

{19 536, 27 436}

In[∘]:=
```
12 965 + 19 536 + 27 436
```

Out[∘]=

59 937

In[∘]:=
```
Monitor[Timing[Tallyρ14 = Tally[Last /@ Tally@Table[ρ[K], {K, AllKnots[{3, 14}]}]]], K]
```

KnotTheory: Loading precomputed data in KnotTheory/14A.dts.

KnotTheory: Loading precomputed data in KnotTheory/14N.dts.

Out[∘]=
```
{207320., {{1, 48336}, {2, 4814}, {3, 217}, {4, 291}, {6, 19}, {5, 4}, {8, 3}}}
```

In[∘]:=
```
Monitor[Timing[TallyHKh14 = Tally[Last /@
    Tally@Table[{Kh[PD@K][q, t], HOMFLYPT[PD@K][a, z]}, {K, AllKnots[{3, 14}]}]]], K]
```

Out[∘]=
```
{6727.34, {{1, 40661}, {2, 6969}, {3, 965},
  {5, 85}, {4, 411}, {6, 43}, {8, 6}, {10, 1}, {9, 1}, {7, 7}}}
```

In[∘]:=
```
{Total[Times @@@ Rest[Tallyρ14]], Total[Times @@@ Rest[TallyHKh14]]}
```

Out[∘]=
```
{11601, 19276}
```

In[∘]:=
```
Total[Times @@@ {{1, 40661}, {2, 6969}, {3, 965},
   {5, 85}, {4, 411}, {6, 43}, {8, 6}, {10, 1}, {9, 1}, {7, 7}}]
```

Out[∘]=
```
59937
```

In[∘]:=
```
Total[Last @@@ {{1, 48336}, {2, 4814}, {3, 217}, {4, 291}, {6, 19}, {5, 4}, {8, 3}}]
```

Out[∘]=
```
53684
```

In[∘]:=
```
59937 - 53684
```

Out[∘]=
```
6253
```

In[∘]:=
```
Total[Last @@@ {{1, 40661}, {2, 6969}, {3, 965},
   {5, 85}, {4, 411}, {6, 43}, {8, 6}, {10, 1}, {9, 1}, {7, 7}}]
```

Out[∘]=
```
49149
```

In[∘]:=
```
59937 - 49149
```

Out[∘]=
```
10788
```

In[∘]:=
```
Monitor[
 {n = 13;
  "All" → NumberOfKnots[{3, n}],
  "ρ" → Length@Union@Table[ρ[K], {K, AllKnots[{3, n}]}],
  "HOMFLY+Kauffman+Kh" →
   Length@Union@Table[{HOMFLYPT[K], Kh[K], Kauffman[K]}, {K, AllKnots[{3, n}]}],
  "HOMFLY+Kauffman" →
   Length@Union@Table[{HOMFLYPT[K], Kauffman[K]}, {K, AllKnots[{3, n}]}],
  "Kauffman" → Length@Union@Table[{Kauffman[K]}, {K, AllKnots[{3, n}]}],
  "HOMFLY+Kh" → Length@Union@Table[{HOMFLYPT[K], Kh[K]}, {K, AllKnots[{3, n}]}]
 }, K]
```

Out[∘]=
```
{All → 12 965, ρ → 12 006, HOMFLY+Kauffman+Kh → 12 010,
 HOMFLY+Kauffman → 12 008, Kauffman → 11 968, HOMFLY+Kh → 11 285}
```